

RootsOfUnity package

A subpackage for QuantumGroups v2.
 Version 2.0, September 18, 2008, Scott Morrison

Introduction

Implementation

```

BeginPackage["QuantumGroups`RootsOfUnity`",
  {"QuantumGroups`", "QuantumGroups`RootSystems`"}];

AlcoveDefiningRoot;
WeightInAlcoveQ;
AlcoveWeights;
AlcoveWeightsInLattice;
AlcoveRoots;
LevelFromRoot;
RootFromLevel;

Begin["`Private`"];

AlcoveDefiningRoot[r_, L_] :=
  AlcoveDefiningRoot[r, L] = With[{lp = If[EvenQ[L], L/2, L]},
    If[Divisible[lp, LacingNumber[r]],
      LongDominantRoots[r][[1]],
      ShortDominantRoots[r][[1]]
    ]
  ]

WeightInAlcoveQ[r_, L_Integer][λ: {__Integer}] := (And @@ (NonNegative /@ λ)) ^
  (KillingForm[r][AlcoveDefiningRoot[r, L], λ + ρ[r]] < If[EvenQ[L], L/2, L])
    
```

```

AlcoveWeights[r_, l_] := AlcoveWeights[r, l] =
Module[{ar = AlcoveDefiningRoot[r, l], λ = ZeroVector[Rank[r]], p},
  Reap[
    While[
      Sow[λ];
      λ += UnitVector[Rank[r], 1];
      While[¬ ZeroVectorQ[λ] ∧ ¬ WeightInAlcoveQ[r, l][λ],
        (* odometer *)
        p = Position[λ, x_ /; x > 0][[1, 1]];
        λ[[p]] = 0;
        If[p < Rank[r], ++λ[[p + 1]]];
      ];
      ¬ ZeroVectorQ[λ]
    ]
  ][[2, 1]]
]

```

```

AlcoveWeightsInLattice[r_, l_, All] := AlcoveWeights[r, l]

```

```

AlcoveWeightsInLattice[r_, l_, lattice_] := AlcoveWeightsInLattice[r, l, lattice] =
Cases[AlcoveWeights[r, l], λ_ /; WeightInLatticeQ[r, λ, lattice]]

```

```

AlcoveRoots[r_, l_] :=
AlcoveRoots[r, l] = Cases[AlcoveWeights[r, l], λ_ /; RootWeightQ[r, λ]]

```

```

LevelFromRoot[r_, l_] :=  $\frac{l}{2 \text{LacingNumber}[r]}$  - DualCoxeterNumber[r]

```

```

RootFromLevel[r_, k_] := 2 LacingNumber[r] (k + DualCoxeterNumber[r])

```

```

AlcoveRoots[r_, l_] :=
AlcoveRoots[r, l] = Cases[AlcoveWeights[r, l], λ_ /; RootWeightQ[r, λ]]

```

```

TwistFactor[r_, Irrep[r_][λ_]] := Global`qKillingForm[r][λ, λ + 2ρ[r]]

```

```

End[];

```

```

EndPackage[];

```